Time Series Calculations

Question: Using the following data, calculate the discrete yearly returns of the stocks, as well as the arithmetic average return, standard deviation, variance, covariance and correlation of their returns.

Date	Adjusted Closing Price ($)	
	CBA	BHP
4/1/2011	52.46	44.25
4/1/2010	51.47	38.9
2/1/2009	26.01	30.11
2/1/2008	47.77	36.31
1/1/2007	48.39	25.17

Time Series Calculations – MS Excel

	A	B	C
1	Date	Return (p.a.)	Return (p.a.)
2		CBA	BHP
3	4/1/2011		
4	4/1/2010	0.0192	0.1375
5	2/1/2009	0.9789	0.2919
6	2/1/2008	-0.4555	-0.1708
7	1/1/2007	-0.0128	0.4426
8			
9	average	=average(B4:B7)	=average(C4:C7)
10	var	=var(B4:B7)	=var(C4:C7)
11	stdev	=stdev(B4:B7)	=var(C4:C7)^(1/2)
12	covar	=covar(B4:B7, C4:C7)*4/(4-1)	
13	correl	=correl(B4:B7, C4:C7)	
14			

Multi-stock Portfolio Variance

Question: Assume there are 3 stocks with the following variances and covariances:

$\sigma_1^2 = 0.01$

$\sigma_2^2 = 0.04$

$\sigma_3^2 = 0.09$

$\sigma_{1,2} = 0.01$

$\sigma_{1,3} = 0.015$

$\sigma_{2,3} = 0.03$

Find the variance of an equi-weighted portfolio of these 3 stocks.

	A	B	C	D
1	weights	=1/3	=1/3	=1/3
2	=B1	0.01	0.01	0.015
3	=C1	=C2	0.04	0.03
4	=D1	=D2	=D3	0.09
5	**Sumproduct method**			
6		=sumproduct(B2:B4,$A$2:$A$4)	=sumproduct(C2:C4,$A$2:$A$4)	=sumproduct(D2:D4,$A$2:$A$4)
7	var(P)	=sumproduct(B6:D6,B1:D1)	0.027777778	
8	**Matrix method**			
9	var(P)	=mmult(mmult(B1:D1, B2:D4), A2:A4)	0.027777778	
10	Note: you must press ctrl+shift+enter after			
11	typing the matrix formula in B9.			
12				

MS Excel: Tangency Portfolio

Question:

Find the tangency portfolio T by investing in the 3 stocks whose returns and standard deviations are given in the below graph. The return of the risk free rate is also given in the graph. Additionally, the correlations between the stocks are all 0.5.



Answer:

Graphically, we are trying to find the weights in stocks 1, 2 and 3 which give the tangency portfolio T. We'll do this by maximizing the gradient of the CAL. The gradient is the Sharpe ratio.



The maximization part requires the MS Excel add-in 'solver'.

First enter the below cells:

	A	B	C	D
1	stock	1	2	3
2	stock return	0.1	0.2	0.3
3	weight in P	=1/3	=1/3	=1/3
4				
5	var-cov matrix	0.04	0.02	0.02
6		=C5	0.04	0.02
7		=D5	=D6	0.04
8				
9	rf	0.05		
10	var(P)	=MMULT(MMULT(B3:D3, B5:D7), TRANSPOSE(B3:D3))		
11	stdev(P)	=B10^0.5		
12	return(P)	=MMULT(B2:D2, TRANSPOSE(B3:D3))		
13	sum of weights	=SUM(B3:D3)		
14	Sharpe ratio(P)	=(B12-B9)/B11		
15	Note: you must press ctrl+shift+enter after			
16	typing the matrix formulas in B10 and B12.			
17				

Now we have to install the solver add-in.

- Click the ‘File’ button in the top left of MS Excel, and a menu should pop up.
- Click ‘Options’ in the bottom of the menu, and a window should pop up.
- Click ‘Add-Ins’ on the left margin of the menu.
- Click on the ‘Solver Add-in’ and then click on the ‘Go’ button. Don’t click OK, that does nothing!
- Another menu will pop up. Make sure ‘Solver Add-in’ is ticked, then click OK.
- Click on the ‘Data’ tab at the top of the excel window.
- You should now see a ‘Solver’ button on the right.

- Click the solver button and fill out the solver menu as follows:



- What we're trying to do is maximize the Sharpe ratio (B14) by changing the weights (B3:D3), while keeping the sum of the weights equal to one (B13).
- Click solve, and hopefully you get the following results, which gives us the weights in the **tangency portfolio**, as well as its return and standard deviation:

	A	B	C	D
1	stock	1	2	3
2	stock return	0.1	0.2	0.3
3	weight in P	-0.555570888	0.3333	1.2222
4				
5	var-cov matrix	0.04	0.02	0.02
6		0.02	0.04	0.02
7		0.02	0.02	0.04
8				
9	rf	0.05		
10	var(P)	0.058273088		
11	stdev(P)	0.241398194		
12	return(P)	0.377781949		
13	sum of weights	1		
14	Sharpe ratio(P)	1.357847561		
15	Note: you must press ctrl+shift+enter after			
16	typing the matrix formulas in B10 and B12.			
17				

Note that portfolio P here is actually the tangency portfolio T.